
TIPO DI DATO ARRAY

Voglio presentarvi alcune situazioni ‘problematiche’ per evidenziare l’inadeguatezza degli strumenti da programmatore che sono attualmente in vostro possesso.

Caso di studio 1: a proposito di medie …

Avete fatto carriera e vi ritrovate prof. Informatica! Avete anche appena terminato di correggere il primo pacco di compiti (sigh) e vorreste contare quanti alunni hanno preso un voto superiore alla media della classe: che informatici sareste se non pensaste di utilizzare un programmino per calcolare una volta per tutte questo valore ?

Program medie;

var 

  i, voto: integer; (* ‘i’ serve per i cicli, ‘voto’ per leggere da tastiera il voto *) 

  somma_voti, num_alunni: integer; (* num_alunni: numero degli alunni *)

  quanti_sopra_media: integer;

  media: real;

begin

  somma_voti:=0; media:=0; quanti_sopra_media:=0;

  writeln(‘Quanti alunni ci sono? ‘);

  readln(num_alunni);

  for i:=1 to num_alunni do

  begin

    writeln(‘Inserisci voto prossimo alunno: ‘);

    readln(voto);

    somma_voti:=somma_voti +  voto

  end;

  media:=somma_voti /num_alunni;

  Dopo aver impiegato solo pochi millesimi di secondo per scrivere questo codice, vi accorgete di … non sapere come andare avanti! Infatti ora che avete calcolato la media dovreste contare i voti che la superano, ma non avete più i voti a disposizione per confrontarli ! Infatti la variabile voto conterrà solo l’ultimo voto letto. I precedenti sono stati sommati nella variabile somma_voti e poi soprascritti ciascuno dal successivo (questo ovviamente perché la readln utilizza nel ciclo sempre la stessa variabile per leggere i voti).

( ( Una prima orripilante soluzione (sicuri di esservi laureati in Informatica??) potrebbe essere quella di chiedere una seconda volta all’operatore gli stessi voti per contare quelli sopra la media 

(si immagina di continuare il programma di prima):

  for i:=1 to num_alunni do

  begin

    writeln(‘Inserisci voto prossimo alunno: ‘);

    readln(voto);

   if voto>media then 

    quanti_sopra_media:= quanti_sopra_media + 1

  end;

Stiamo scherzando? A parte le furiose proteste dell’utente che si vede costretto a ripetere l’inserimento dei dati, provate a pensare al rischio di inserire dati diversi dai precedenti con risultati del tutto falsati…

ATTENZIONE: la seconda ‘soluzione’ che viene proposta qui di seguito è, come il primo, solo un esempio. Come tale è NECESSARIO per apprezzare e capire l’uso del nuovo strumento di programmazione che presenterò, ma non deve essere ‘studiato’.

( ( Seconda soluzione (lascio decidere a voi se più o meno orripilante della precedente): ‘banale, come ho fatto a non pensarci prima: basta usare una variabile diversa per ogni voto!’

Program medie;

var 

  i, somma_voti, num_alunni, quanti_sopra_media: integer;

  media: real;

voto1, voto2, voto3, voto4, …, voto32 (aiuto!): integer;

begin

  soma_voti:=0; media:=0; quanti_sopra_media:=0;

  writeln(‘Quanti alunni ci sono? ‘);

  readln(num_alunni);

Il bello arriva adesso! Poiché non si sta usando una sola variabile (voto) per leggere i voti, non è possibile usare un ciclo per leggere i voti! E’ necessario leggere il primo. Poi se il numero di alunni è maggiore di 1, significa che devo sicuramente leggerne anche un secondo; poi se il numero di alunni è anche maggiore di due significa che devo leggerne almeno anche un terzo e così via …

                     writeln(‘Inserisci voto primo alunno: ‘);

                     readln(voto1);

                     somma_voti:= somma_voti + voto1

                   if num_alunni>1 then 

                     begin

                      writeln(‘Inserisci voto secondo alunno: ‘);

                      readln(voto2);

                      somma_voti:= somma_voti + voto2
                     end;

                     if num_alunni>2 then 

                     begin

                      writeln(‘Inserisci voto prossimo alunno: ‘);

                      readln(voto3);

                      somma_voti:= somma_voti + voto3
                     end;

e così via fino a … 

                    if num_alunni>31 then 

                     begin

                      writeln(‘Inserisci voto 32-mo alunno: ‘);

                      readln(voto32);

                      somma_voti:= somma_voti + voto
                     end;

  media:=somma_voti / num_alunni;  (* alleluia … *)

e ora … ripetiamo tutto per contare quelli sopra la media

                     if voto1>media then

                       quanti_sopra_media:= quanti_sopra_media + 1;

                     if num_alunni>1 then 

                        if voto2>media then

                         quanti_sopra_media:= quanti_sopra_media + 1;

                       ecc. ecc. (abbiate pietà !)

Non so voi, ma io ho già perso di vista l’obiettivo che ci eravamo posti… Immaginate se invece degli alunni di una sola classe considerassimo quelli dell’intero istituto per una qualche statistica!

ED ECCO LA SOLUZIONE …

Quello che ci serve è un contenitore (si tratterà poi pur sempre di una variabile, anche se di tipo particolare) capace di contenere non un solo voto ma un insieme di voti. Il nome del contenitore in questo modo sarebbe uno solo, risolvendo l’inconveniente del proliferare selvaggio del numero delle variabili; inoltre, come vi spiegherò tra poco, sarà ancora possibile usare i cicli per elaborare con poche righe di codice tutti i dati. 

VOTI

	7
	4
	5
	6
	6
	8
	8
	5


   voti[1]          voti[2]        voti[3]          voti[4]           voti[5]         voti[6]            ecc.       

· Il nome del contenitore è VOTI. Questo è il nome che il programmatore ha deciso di dare all’ARRAY. Quando l’array è come quello nel disegno soprastante (lineare, monodimensionale) è anche chiamato vettore. Esistono array bidimensionali (pensate ad una griglia tipo battaglia navale, o ad uno schema tipo parole crociate) chiamati matrici.


· I dati non sono depositati nel vettore alla rinfusa ma tenuti in ‘scomparti’ separati.


· Ogni ‘scomparto’ occupa una precisa posizione (1, 2, 3, 4 ecc.)  specificata tramite un numero detto indice che va indicato tra parentesi quadrate dopo il nome dell’array come in voti[1], voti[2] ecc.

Ecco come si dichiara in Pascal una variabile array adatta a contenere 32 interi:

var                                                      (* NOTA. sarebbe meglio mettere una costante invece del 32:

  voti: array[1 .. 32] of integer;                          … array[1 .. NUM_ALUNNI] of integer *) 

Per poter passare gli array ai sottoprogrammi è però meglio abituarsi fin da subito a dichiarare prima un  nuovo tipo di dati e solo dopo le variabili di quel tipo:

NOTA: la sezione type deve essere scritta prima della sezione var

type                                      

   vet_int = array[1 .. 32] of integer;  (* vet_int sta per ‘vettore di interi’ *)

var 

  voti: vet_int;

OSSERVAZIONI

· Quando si dichiara un tipo non si usano i due punti ma l’uguale (… = array … )

· Nella sezione var è poi possibile dichiarare tutte le variabili di quel tipo che si vogliono: tutte corrisponderanno a contenitori per 32 interi

Voti è di tipo vet_int, cioè un array[1 .. 32] of integer, esattamente come prima. Anche un parametro di un sottoprogramma potrà essere dichiarato di tipo vet_int (la dicitura array[1 ..32] of integer non sarebbe infatti accettata come tipo di un parametro).

Una volta specificata la posizione tra parentesi quadre la scrittura indica esattamente una variabile del tipo contenuto nel vettore. Detto con altre parole: vet[7] è, a tutti gli effetti, una variabile integer e può essere usata dove siete abituati ad usare una qualsiasi altra variabile integer:

Writeln(‘Dimmi un valore e lo memorizzerò nella posizione 4 del vettore’);

readln( voti[4] );


writeln(‘Ora visualizzo il contenuto del sesto elemento del vettore: ‘);

writeln( voti[6] );  


if vet[2]>x then … 


repeat 

  …

until x>vet[5];


x:=6 * vet[3] – 2*vet[7]


writeln ('Dimmi una posizione e visualizzerò il contenuto di quell’elemento: ‘);

readln (posizione);

writeln ( voti[posizione] );

L’ultimo esempio è particolarmente interessante: invece di specificare un numero preciso, come indice è possibile mettere una variabile integer. Questo dà al programmatore la possibilità di scandire, passare in rassegna, più elementi del vettore cambiando in un ciclo il valore della variabile indice:

for i:=1 to num_voti do

  writeln( voti[i] );
Cercate di capire bene questo spezzone di codice: è alla base di tutte le elaborazioni con gli array che vi saranno spiegate in questo corso!

La variabile i, grazie al ciclo, assume tutte le posizioni del vettore e di volta in volta individua nel ciclo elementi successivi, uno dopo l’altro dal primo all’ultimo. La prima volta che viene eseguito il ciclo la i vale 1 e quindi la writeln corrisponde a writeln(voti[1]) e viene quindi visualizzato il contenuto del primo elemento del vettore. Poi la i diventa 2, e viene elaborato il secondo elemento e così via fino all’ultimo indicato nel ciclo (num_voti).

E’ arrivato il momento di mettere tutto insieme e risolvere elegantemente il problema di partenza …

Program medie;

const
  MAX_ALUNNI=32;

type 

  vettore_32_interi = array[1 .. MAX_ALUNNI] of integer;

var 

  voti: vettore_32_interi;

  i, somma_voti, num_alunni, quanti_sopra_media: integer;

  media: real;

                   begin

  somma_voti:=0; media:=0; quanti_sopra_media:=0;

  writeln(‘Quanti alunni ci sono? ‘);

  readln(num_alunni);

(* CARICAMENTO DEI DATI NEL VETTORE *)

  for i:=1 to num_alunni do

  begin

    write(‘Inserisci voto per l’’ alunno ‘,i,’: ‘);

    readln(voti[i]);

    somma_voti:=somma_voti +  voti[i]

  end;

  media:=somma_voti / num_alunni;

(* I DATI SONO ANCORA NEL VETTORE: CONTROLLIAMO TUTTI *)  

for i:=1 to num_alunni do

                     if voti[i]>media then

    quanti_sopra_media:= quanti_sopra_media + 1;

writeln(‘Risultato: ‘,quanti_sopra_media);

readln;

                end. 



Caso di studio 2: MENU

Attraverso questo esempio imparerete a passare i vettori ai sottoprogrammi.  

Un menu presenta sullo schermo un elenco di opzioni delle quali una sola sarà quella scelta dall’utente. Un menu può presentarsi a tutto schermo, come nell’esempio qui sotto a sinistra, o sotto forma di barra dei menu, come nell’esempio qui sotto a destra.





Chiunque abbia provato ad usare un programma per computer sa che sono veramente tante le situazioni in cui è necessario dialogare con l’utente presentando un menu di scelte. 

Limitandoci alla tipologia rappresentata qui sopra a sinistra, fondamentalmente lo scopo di un menu è sempre lo stesso: scrivere un titolo, sotto di questo scrivere un certo numero di scelte e leggere la risposta dell’utente. Situazione perfetta per la scrittura di un sottoprogramma: comunico titolo, opzioni, numero di queste ultime e come valore di ritorno ci si aspetterà il numero della scelta fatta. Naturalmente il sottoprogramma provvederà anche a rifiutare scelte impossibili (valori minori di 1 o più grandi del numero delle opzioni previste), a visualizzare messaggi di errore e a far ripetere la scelta fino a che l’utente non indicherà forzatamente una delle opzioni previste:












La chiamata alla funzione menu che corrisponde all’esempio cui ci stiamo riferendo sarebbe allora:

scelta:=menu(‘Scegli il gioco che preferisci’,4,’Tennis’,’Poker’,’Scacchi’,’ESCI’)

Un’altra situazione tipica è quella in cui, segnalando un errore, si vuole chiedere all’utente come procedere. Immaginiamo che si sia verificato un errore di registrazione di un documento:

scelta:=menu(‘Errore di scrittura su disco’,3,’Riprova’,’Ignora’,’annulla’)

Scusi prof., mi sa che ha dimenticato un parametro nel secondo esempio (lo studente medio finge clamorosamente di essere disinteressato a quello che scrivi, ma prova a sbagliare una virgola …): devono essere sei (il titolo, numero opzioni e le quattro scelte).

E già, ma se le scelte sono solo 3 come nel secondo esempio? O se ne servissero 12?? L’unica soluzione ora alla vostra portata sarebbe quella di prevedere il massimo numero possibile di scelte, ignorando quelle che di volta in volta non servono; la dichiarazione dell’intestazione di questa funzione sarebbe un vero ‘mostro’ che potrebbe assomigliare alla seguente:

Function menu(titolo: string; num_opzioni: integer; 

                        scelta1,scelta2,scelta3,scelta4,…,sceltaN: string): integer;

dove N rappresenta il massimo numero di scelte utilizzabile; se le opzioni fossero di più la situazione non sarebbe gestibile a meno di modificare il sottoprogramma.

Chiamare questa funzione sarebbe veramente una ‘pena’. Immaginiamo infatti di voler chiedere una conferma all’utente (in questo caso le opzioni sarebbero solo due: sì o no):

scelta:=menu(‘Sei sicuro’,2,’Si’,’No’,’’,’’,’’,’’,’’,’’,’’…’’)

anche se le opzioni sono solo due, TUTTI i parametri devono comunque essere sempre indicati (qui si immagina che quelli inutili siano indicati come stringa nulla (‘’)

 Anche il codice della funzione sarebbe piuttosto pesante ed intricato:

Function menu(titolo: string; num_opzioni: integer; 

                        scelta1,scelta2,scelta3,scelta4,…,sceltaN: string): integer;

begin

   writeln(titolo);  (* e fino a qui …*)

e qui immaginatevi, un po’ come nel caso della media, tutta una serie di if … then che a seconda del numero di opzioni specificato vanno o non vanno a stampare la variabile stringa che corrisponde a quell’opzione; per semplicità mi limito solo ad iniziare …

writeln(‘1 – ‘,scelta1);

if num_opzioni>1 then

  writeln(‘2 – ‘,scelta2);

if num_opzioni>2 then

  writeln(‘3 – ‘,scelta3);

ecc. 

ED ECCO LA SOLUZIONE CON GLI ARRAY …

La soluzione che sfrutta gli array prevede invece di passare le opzioni sotto forma di vettore di stringhe:

             type vettore_stringhe = array[1 .. 20] of string;

Function menu(titolo: string; num_opzioni: integer; scelte: vettore_stringhe): integer;

var i,scelta: integer; 

begin

              writeln(titolo);  

             for I:=1 to num_opzioni do

               writeln(I, ‘ – ‘, scelte[i]);

           writeln(‘Scegli -> ‘);

           readln(scelta);

           menu:=scelta

          end;

OPERAZIONI TIPICHE CON GLI ARRAY

In tutti gli esempi che seguono si immaginano valere le seguenti dichiarazioni:

const
  MAX_ELEMENTI=100;

type 

  vettore_interi = array[1 .. MAX_ELEMENTI] of integer;

var 

  vettore: vettore_interi; N: integer; (* N è il numero di elementi da usare *)

1. Caricamento di un vettore di N elementi, con N < MAX_ELEMENTI


for i:=1 to N do

begin

  writeln(‘Inserire elemento n. ‘, i)

  readln( vettore[i] )

end;

Vediamo la stessa operazione fatta con un sottoprogramma. Poiché non intendiamo avvantaggiarci di un valore restituito da una funzione, useremo una procedura. 

Essa riceverà il vettore da caricare per indirizzo: diversamente non potremmo modificare in modo permanente i valori degli elementi del vettore passato alla procedura. Il secondo parametro corrisponde al numero di elementi da caricare (chi usa la procedura deve assicurarsi di non passare come numero di elementi un valore superiore alla capacità massima del vettore).

procedure carica_vettore(VAR vet: vettore_interi; N: integer);

var i: integer;

begin

  for i:=1 to N do

  begin

    writeln(‘Inserire elemento n. ‘, i);
    readln( vet[i] )

end;

end;
NOTA

Qualcuno potrebbe obiettare che la procedura risulta costituita da un numero di righe di codice superiore rispetto alla soluzione che non usa i sottoprogrammi. Ricordiamoci però che la procedura potrà essere chiamata tutte le volte che si vuole per caricare vettori dello stesso tipo (ANCHE in programmi diversi: basta copiarla/incollarla od includerla in una libreria). Qualche volta si tratterà di voti, altre volte di temperature, pesi od altezze ecc. La procedura non conosce il significato degli interi che sta caricando: svolge in modo anonimo il suo compito: riempire di interi un vettore che riceve come parametro, punto.

Suggerimento: prova a rendere ancora più generale la procedura prevedendo un terzo parametro corrispondente ad un messaggio personalizzato che solleciti l’utente in modo appropriato. In questo modo invece del generico ‘Inserire elemento n.’, potranno essere usati messaggi tipo ‘Inserire la temperatura n.’, o ‘Inserire il peso dell’oggetto n.’ ecc.

2. Visualizzazione degli elementi di un vettore di N elementi, con N < MAX_ELEMENTI


for i:=1 to N do

  writeln( vettore[i] );

3. Somma dei valori degli elementi di un vettore di N elementi, con N < MAX_ELEMENTI


somma:=0;

for i:=1 to N do

  somma:=somma + vettore[i] ;

4. Media degli elementi di un vettore di N elementi, con N < MAX_ELEMENTI

NOTA: questo esempio è particolarmente interessante se utilizziamo i sottoprogrammi. Infatti la funzione che calcola la media può richiamare quella, già scritta, che calcola la somma!


somma:=0;

for i:=1 to N do

  somma:=somma + vettore[i] ;
media:= somma / N;

5. Posizione dell’elemento con valore minimo in  un vettore di N elementi, con N < MAX_ELEMENTI


L’algoritmo è basato su questo ragionamento: all’inizio ipotizzo che il minimo sia l’elemento che si trova in posizione 1. Quindi controllo tutti gli altri elementi, dal secondo in poi, e tutte le volte che trovo una posizione con un valore più piccolo diventa quest’ultima la nuova posizione del minimo.

posizione_min:= 1;

for i:=2 to N do

 if  vettore[i] < vettore[posizione_min] then   (* trovata posizione i con valore più piccolo *)
   posizione_min := i;

Funzione per la stessa operazione.

function posizione_min(vet: vettore_interi; N: integer):integer;

var pos_min:integer;

begin 

  pos_min:= 1;

  for i:=2 to N do

  if  vet[i] < vet[pos_min] then   (* trovata posizione i con valore più piccolo *)
   pos_min := i;

  posizione_min := pos_min

end;

NOTA: se nel vettore ci sono più elementi minimi con lo stesso valore, viene individuato il primo (quello con indice minore). Infatti quando viene confrontato il primo minimo con gli altri la condizione vet[i] < vet[pos_min] non è verificata e l’aggiornamento della posizione non viene effettuata. Aggiungendo l’uguaglianza al confronto (vet[i] <= vet[pos_min]) verrebbe invece localizzato l’ultimo minimo (quello con indice superiore).

6. Posizione dell’elemento con valore massimo in un vettore di N elementi, con N < MAX_ELEMENTI


E’ sufficiente cambiare al punto precedente posizione_min con posizione_max, < con > e pos_min con pos_max per ottenere l’algoritmo desiderato.

7. Ricerca della posizione di un valore in un vettore di N elementi, con N < MAX_ELEMENTI



writeln(‘Quale elemento cerchi? ‘);

readln(cercato);

posizione:=0;

for i:=1 to N do

  if vettore[i] = cercato then

     posizione:=i;

function cerca_vettore(vet: vettore_interi; N: integer; cercato: integer):integer;

var posizione: integer;

begin 

  posizione:=0;

  for i:=1 to N do

    if vet[i] = cercato then

       posizione:=i;

cerca_vettore:=posizione

end;

Ed ecco la soluzione con il repeat, nel caso si voglia il primo elemento uguale a quello che si sta cercando. Questa soluzione può essere anche molto più efficiente di quella con il for. Ad esempio se gli elementi del vettore fossero diecimila e quello che si sta cercando fosse nella quarta posizione, il ciclo for andrebbe inutilmente a controllare gli altri 9996, mentre il ciclo repeat si fermerebbe al quarto!

posizione:=0; i:=0;

repeat

  i:=i+1;

  if vettore[i] = cercato then

     posizione:=i

until (i=N) or (posizione<>0);

7. Ordinamento dei valori in un vettore di N elementi, con N < MAX_ELEMENTI

    METODO PER SCAMBIO – BUBBLE SORT


Quello che sto per presentare è un metodo non molto efficiente quando il numero degli elementi supera il centinaio, ma ha il pregio di essere breve, relativamente semplice ed abbastanza facile da ricordare. Appartiene alla famiglia degli algoritmi di ordinamento (SORT) detti per scambio perché adotta una scansione sistematica che confronta ‘ciecamente’ tutti gli elementi e scambia quelli fuori posto tra loro. Lo scambio di elementi è usato in modo pesante: altri algoritmi adottano tecniche che ‘ragionano’ di più sulla situazione del vettore od adottano tecniche più intelligenti (ripagate solo se il numero degli elementi è sufficientemente elevato) diminuendo il numero di scambi necessari. Ovviamente questi algoritmi sono più difficili …

Prima di esaminare il codice Pascal conviene presentare l’algoritmo in modo ‘informale’ aiutandosi anche con una simulazione grafica. Come punto di partenza consideriamo il seguente vettore disordinato:


	9
	8
	2
	6
	3


L’algoritmo procede in questo modo:

1. confronta la prima posizione con la seconda: se il numero contenuto nella prima posizione è maggiore di quello nella seconda scambia i numeri:






NOTA: tenete d’occhio il numero 9; come una bolla d’aria nell’acqua  risalirà verso la ‘superficie’ ( la parte finale del vettore), da cui il nome dell’algoritmo (bubble=bolla sort=ordinamento)


2. spostati di una posizione verso destra e ripeti il confronto/scambio 





3. ripeti il punto 2 fino a raggiungere la penultima posizione (che verrà confrontata con l’ultima)











Dopo questo primo ‘giro’ (passata) l’elemento più grande del vettore (il 9) ha raggiunto la sua giusta posizione. Ora si ripete il tutto fermandoci però alla penultima posizione (l’ultima, come detto prima, è già ordinata e va lasciata stare!). Per ricordarcelo, sfumiamo in grigio l’ultima casella del vettore.









Ora ripetiamo ma fermandoci alla terzultima posizione (le ultime due sono in ordine)




            nessuno scambio 



NOTA: il vettore sarebbe già in ordine, ma l’algoritmo ‘non se ne accorge’ e, ciecamente, continua i confronti:




nessuno scambio 


l’algoritmo si ferma quando ha ripetuto il processo n-1 volte

Ed ecco la codifica in pascal:

(* per n – 1 volte … *)
for i:=1 to n - 1 do         

(* parti dalla prima posizione e confronta 

   tutte le coppie fino alla n – i *)

  for j := 1 to n - i do      

(* se trovi due celle adiacenti non 

  in ordine, scambiale *)

      if vettore[j] > vettore[j+1] then         

     begin

         tmp:=vettore[j];

         vettore[j]:=vettore[j+1];

         vettore[j+1]:=tmp

      end;  

7. Ordinamento dei valori in un vettore di N elementi, con N < MAX_ELEMENTI

    METODO PER SELEZIONE – PER RICERCA DEL MINIMO


Questo algoritmo appartiene alla categoria chiamata ‘per selezione’ perché invece di scambiare in continuazione celle adiacenti tra loro, cerca il miglior numero da spostare, riducendo di molto il numero degli scambi. In dettaglio:

1. Partendo dalla prima posizione, localizza la cella con valore minimo; prendi il minimo e mettilo in prima posizione; il numero in prima posizione mettilo nella cella in cui hai trovato il minimo:

cerca il minimo                                            scambio                                         risultato




	9
	8
	2
	6
	3





2. Spostati in avanti di una casella ed ignora la prima (già in ordine): cerca il minimo nella parte del vettore ‘avanzata’; mettilo in seconda posizione; il numero in seconda posizione mettilo nella cella in cui hai trovato il minimo. Ripeti il procedimento fino alla penultima posizione (a quel punto il numero in ultima posizione è per forza in ordine.




Ed ora il codice pascal:

for i:=1 to n-1 do

begin

  pos_min:=i;

(* cerco la posizione del minimo

    tra i e ultimo elemento *)  

  for j:=i+1 to n do

     if vet[j] < vet[pos_min] then

        pos_min:=j;

   (* scambio con il minimo *)

   tmp := vet[i] ;

   vet[i] := vet[pos_min];

  vet[pos_min]:=tmp 

end;

Scegli il gioco che preferisci





Tennis


Poker


Scacchi


ESCI





SCEGLI  => _





File





Modifica

     Salva





     Ecc.














MENU





Titolo 





numero della scelta fatta 





opzione 1





opzione 2





opzione 3





opzione 4





n. opzioni





OSSERVAZIONI





Se gli alunni diventassero anche 10000 il codice rimarrebbe identico! Sarebbe sufficiente modificare il valore della costante MAX_ALUNNI�


Non è obbligatorio usare tutte le posizioni del vettore: al massimo ce ne sono disponibili MAX_ALUNNI ma il valore comunicato al computer (num_alunni) potrebbe anche essere inferiore. A dire la verità, il programma dovrebbe controllare che questo valore sia compreso tra 1 e MAX_ALUNNI perché��tentare di usare un elemento prima dell’inizio del vettore o dopo la fine è un gravissimo errore che spesso porta al blocco del programma ed eventualmente del computer stesso�








Stampa tutte le opzioni precedute dalla I e da un trattino: 1 – opzione1, 2 – opzione 2 ecc.





Per semplicità non è stato controllato l’input dell’utente (che potrebbe essere <1 o >num_opzioni)





NOTA: caricare un vettore di 


stringhe, piuttosto che di reali


comporterebbe solo la modifica


delle dichiarazioni nella sezione


type.








Procedura per visualizzare gli elementi di un vettore. Riceve il vettore (per valore, visto che non dobbiamo modificare in alcun modo il suo contenuto!) ed il numero di elementi da considerare.





NOTA: chi usa la procedura deve assicurarsi di non passare come numero di elementi un valore superiore alla capacità massima del vettore.





procedure visualizza_vettore(vet: vettore_interi; N: integer);


var i: integer;


begin


  for i:=1 to N do


      writeln( vet[i] )


end;








Funzione per la stessa operazione. 





function  somma_vettore(vet: vettore_interi; N: integer):integer;


var tot: integer;


begin


  tot:=0;


  for i:=1 to N do


      tot:=tot + vet[i] ;


 somma_vettore:=tot


 end;








Funzione per la stessa operazione. 





function  media_vettore(vet: vettore_interi; N: integer):integer;


begin


   media_vettore:= somma_vettore(vet,N) div N


end;





NOTA: un sottoprogramma può passare ad un altro i suoi stessi parametri, con le stesse modalità viste sino ad ora (per valore o per indirizzo).





OSSERVAZIONI





Se l’elemento non viene trovato la variabile posizione rimarrà 0. E’ importante controllarla prima di usarla!  Ad esempio: if posizione>0 then …(* elemento trovato, possiamo procedere *)�


Se nel vettore l’elemento cercato appare più volte, il ciclo for localizza l’ultimo (comunque la scansione arriva ad N); per localizzare il primo è necessario controllare ‘manualmente’ la ricerca usando un ciclo repeat (come mostrato più avanti)











function cerca_vettore(vet: vettore_interi; N: integer; cercato: integer):integer;


var posizione,i: integer;


begin 


  posizione:=0; i:=0;





  repeat


    i:=i+1;


    if vet[i] = cercato then


       posizione:=i


  until (i=N) or (posizione<>0);





cerca_vettore:=posizione


end;

procedure bubble_sort(VAR vet: vettore_interi; n: integer);


var i,j, tmp: integer;


begin


  for i:=1 to n - 1 do         


   for j := 1 to n - i do      


      if vet[j] > vet[j+1] then         


     begin


         tmp:=vet[j];


         vet[j]:=vet[j+1];


         vet[j+1]:=tmp


      end


end;








procedure selezione_sort(VAR vet: vettore_interi; n: integer);


var i,j, tmp, pos_min: integer;


begin


  for i:=1 to n-1 do


  begin


    pos_min:=i;


  


    for j:=i+1 to n do


      if vet[j] < vet[pos_min] then


          pos_min:=j;


  


    tmp := vet[i] ;


    vet[i] := vet[pos_min];


    vet[pos_min]:=tmp 


  end


end;

